

## バリエーションを使い分けた除算器の設計

論理回路を設計していて除算回路が必要になったことがあるという方は、そう少なくないと思います。純粹に数学的な演算もさることながら、デジタル信号処理においても最近ではアルゴリズムが複雑化しており、除算回路の必要性は高まっています(その一方で、除算を行わなくてもすむようなアルゴリズム上のくふうも進歩しているが…)。

除算回路の構成法は、古いものまで含めるとかなりのバリエーションがありますが、意外に知られていないのも事実です。除算回路は「手間のかかる回路」なので、ターゲット・デバイスの処理速度が遅かったところに少しでも速くしようと、かなりくふうが施された歴史があります。もちろん現在でも、進歩の途上にあります。多くのバリエーションをうまく使い分けることが、除算回路のおもしろみではないかと筆者は考えます。

### ●「除算器はかさばる」という先入観を捨ててみる

除算回路というと、「かさばるもの」と考えている方も少なくないようです。しかし、リアルタイム性が必要なく、のんびり演算することが許されるのであれば、分母の数のビット幅をもつ減算器とマルチプレクサ、それに簡単なシーケンサがあれば、分子がどんなに大きいけたの数であっても原理的には演算が可能です。条件によっては、決してかさばるものではありません。例えば、画像信号にはブランキング期間という画像情報のない期間があります。クロックごとに除算結果を出す必要がなく、穏やかなサイクル数(スループット)が期待される場合、そうしたデータの休止期間などを利用してシーケンシャルな除算器をのんびり回す、というような構成手段が選択できれば、回路規模を増加させることなく機能を実現できます。

除算回路は「IP (intellectual property) コアとして提供されるもの」、あるいは「高級な論理合成ツールが吐き出してくれるもの」と割り切っている方もいるかと思います。しかし、除算回路のIPコアを選択する際に、構造として適当なものを選ぶには、除算についての最低限の知識が必要となります。まして、必要な性能の除算器が手に入らないのであれば、自分で何とかするしかありません。

食わず嫌いな方にあえて申し上げるなら、除算回路はそれほど難しいものではありません。わずかな基本事項を理解するだけで、かなりの範囲に応用できるはずです。

## 4-1 まずは筆算のおさらい

筆算による除算というのは小学校で習った割り算の方法で、どなたも覚えがあると思います。図4-1に、その例を示しましょう。

手順は説明するまでもないのですが、以下のとおりです。1回目に152の上位2けたである「15」と除数である「11」を比較し、11の1倍が引けそうなので商に1を立て、 $15 - 11 = 4$ を演算します。2回目には余4を1けたシフト・アップ<sup>注41</sup>し、下位1けたを加えて「42」とします。11の3倍である33が引けそうなので商に3を立て、 $42 - 33 = 9$ を演算します。2回の比較と減算によって除算が構成されていることになります。

### ● 「除算」のとらえかたによって回路構成法も変わる

ここで、用語を以下のように統一します。本連載で各引き数を示す記号もこれに統一します。

- $z$  : 被除数 (dividend)
- $d$  : 除数 (divisor)
- $q$  : 商 (quotient)
- $r$  : 余 (remainder)

これらを用いて引き数の関係を表現すると、式(4-1)が成立します。

$$z = d \times q + r \quad \dots\dots\dots (4-1)$$

そもそも除算とは何でしょうか。図4-1の例では、「152の中に11が何個とれるか」といった考えかたができます。分数で考えれば、分母11と分子152の比と考えることもできます。そして、それぞれの考えかたに対応するように、除算回路の構成法が存在します。筆算による方法はもちろん前者で、その代表格になります。

式(4-1)を変形すると、式(4-2)になります。除算にこだわるなら、こちらの表現のほうが妥当かもしれません。

$$\frac{z-r}{d} = q \quad \dots\dots\dots (4-2)$$




図4-1 筆算による除算  
小学校のときに習った割り算のやりかた。演算回路を作るとい  
う視点から見ると、図の除算は2回の比較と減算によって構成  
されているということになる

注41：シフト・アップとは、除数のけた位置を固定して、被除数(余も含めた)をけた上げしていくことを指す。図4-1では、被除数のけた位置を固定し、除数をシフト・ダウンしていることに注意。

式(4-1)と式(4-2)は同じ関係を表現しているのですが、式(4-1)は乗算、式(4-2)は除算で表現されていることがわかります。つまり、乗算と除算は表裏一体の関係にあるわけです。冒頭で、除算回路は「手間のかかる回路」と述べたのですが、同じ関係にありながらなぜ除算は手間がかかるのでしょうか（「乗算は手間がかからない」という意味ではなく、あくまでも比較の問題）。

乗算は、“sequential addition”と言われます。部分積を求めて加算を繰り返せば答えに到達することができます。これに対して、除算は“sequential subtraction”と言われます。けた単位で減算を行い、そこでの余（部分余；partial remainder, 以下  $pr$  と表記）を求めるという操作を繰り返して真の商に近づけていきます。ここで、引き過ぎてはならないという条件が必要になります。減算であることと引き過ぎてはならないという、二つの制約が除算をめんどうなものにしています（引き過ぎててもよいとする方法もあるが、その場合でも厳密な条件が必要になる。詳細は4-4「引き過ぎててもよい非回復法」を参照）。

見かたを変えて式(4-1)を乗算と加算の複合として演算するとすれば、除数  $d$ 、商  $q$ 、余  $r$  が既知である必要があり、被除数  $z$  は未知数となります。しかし、一般に除算において既知であるのは  $d$ 、 $z$  のみであり、 $q$ 、 $r$  は未知数です。

引き数どうしは同じ関係にありながらアプローチが異なるため、演算の個性も変わってくるといえるのではないのでしょうか。

## 4-2 回復法の定義と成立条件

「除算では減算を繰り返しながら商を求めるが、引き過ぎてはならない」と上述しました。この制約に厳密に従った方法が回復法という回路構成になります。「引けそうであれば引き、引けそうでなければ引かない」というルールの下で、減算とシフト・アップを繰り返します。

### ● 2進法で演算を行うと、“暗黙の乗算”が不要になる

構造的には、「引けそうであるか否か」が比較にあたり、「引く」がそのまま減算にあたります。しかし、「引いてしまってから符号を調べて、負であれば引く前の値に戻す」という手順でも同じ結果が得られるので、ここから回復法の名があるわけです。構造としては、当然、コンパレータと減算器を兼用したものが一般的でしょう。

図4-1の例では10進法(10を法とする)で演算を行いました。論理回路でこれを実現すると、2進法(2を法とする)が基本になります。図4-1をそのまま2進法に変更して同一の演算を行うと、図4-2のようになります。

比較と減算の演算を1サイクルとして、5サイクル繰り返されていることがわかります。なお、図では比較の意味で、除数“1011”を減算できないサイクルにおいても表示しています。また、図の中で「部分余1001をシフト・アップし」とありますが、これは次のサイクルの除数を基準として行っているのです。図ではシフト・アップされているように見えない点に注意してください。操作そのものは、図4-1となんら変わるところはありません。

ここで、2を法として演算することの利点を明確にしておかなくてはなりません。2のべき乗や10など、2より大きな数を法として演算する場合には、逆の作用を及ぼすからです。

図4-1の2回目の演算で「42 - 33」を行っていますが、この33という数は除数11の3倍です。つまり、



図4-2 図4-1の演算を2進法に変換

図4-1では10進法で割り算を行ったが、それをそのまま2進法に置き換えたものを図に示す。操作そのものは、図4-1となら変わらない




図4-3 回復法のアルゴリズム (その1)

図4-2に忠実に比較と減算を分けて考えた場合のアルゴリズム




図4-4 回復法のアルゴリズム (その2)

「引いてしまったから符号を調べて、負であれば引く前の値に戻す」という手順で除算を行うアルゴリズム。比較と減算を兼用した形。繰り返しのループ部分のみ示す

11×3を暗黙のうちに乗算しているわけです。これをそのまま回路化すると、当然乗算器が必要になります。一方、2を法として演算するとき、商として立つのは'0'または'1'です。つまり、減算に必要な数は'0'か除数そのものであり、乗算器は不要です。これが、2を法とする除算を行う場合の最大の利点と言えるでしょう。

さて、ここで回復法のアルゴリズムを明確にしておきましょう(図4-3)。まずは、図4-2に忠実に、比較と減算を分けて考えてみます。

次に、先にも述べたとおり、比較と減算を兼用します。すると、フローチャートは図4-4のように書き換えられます。

### ●「部分余<2×除数」でなくては演算が成立しない

回復法が除算アルゴリズムとして成立するには、条件があります。

図4-2で行った除算において、除数が11ではなく4だとどうなるのでしょうか。けた位置をまったく変えずに行った場合の演算を図4-5に示します。正しくは $152 \div 4 = 38$ (余0)なのですが、図4-5の演算では商が31で余が28となってしまいます。 $31 \times 4 + 28 = 152$ で関係は正しいのですが、除算としては正しくありません。

こうなる原因は、図4-3における初期化にありました。図4-3のアルゴリズムは、 $pr_i \geq d$ の比較を行う段階で $pr_i < 2d$ でなければならないのです。図4-5の例の場合、最初のサイクルで比較される9は除数4の2倍以上なので、ここでの商として2を立てたいところです。しかし、2を法としているかぎり、それはできません。

一般化すると $pr_i \geq 2d$ の場合は $pr_{i+1} = pr_i - d \geq d$ が成立するため、次のステップでシフト・アップ




図4-5 回復法の失敗例

最初のサイクルで比較される9は除数4の2倍以上となっている。図4-3のアルゴリズムでは、部分余<2×除数という条件が成立しないと正しい結果が得られない

と加算を行うと、 $pr_{i+1} \geq 2d$ となり悪循環は続いていきます。

逆に、 $pr_i < 2d$ を満たしておけば $pr_{i+1} = pr_i - d < d$ が成立し、次のステップでも $pr_{i+1} < 2d$ がならず成立することになります。つまり、初期化の段階で $pr_{initial} < 2d$ となるようにけたを合わせておけば、除算は破たんすることなく最後までまっとうされるわけです。

この条件を満たす方法は何通りかあります。被除数の上位に0を拡張し、かならず $pr_{initial} < 2d$ を成立させるというのも一つの手です。図4-5の例では、被除数に0を3ビット分拡張すれば(“0001”となる)、 $d > 0$ の場合は条件を満たせます。この方法の利点は、めんどろな操作はいっさい必要ないことです。しかし、余分に $i$ のループを回さなければなりません。

浮動小数点形式のように、上位側から見た最初の‘1’が現れるけたを左に詰めてしまうのも有効です。 $pr_i$ と $d$ の両方にこの操作を施せば、商に0が立つ初期のむだなサイクルがほとんど排除されます。ただし、左詰めを行うためのハードウェアと、 $pr_i$ と $d$ をシフトした分を最終的な商や余に対して補正する必要があります。いずれも一長一短があるので、目的に応じて使い分ければよいでしょう。

図4-3のアルゴリズムで除算を成立させる条件がもう一つあります。それは、除数と被除数ともに正であることです。符号付きの演算も可能なのですが、このままのアルゴリズムでは不可能です。図4-3のアルゴリズムは、被除数から除数を減算し続けることで余を正側から0に近づけます。しかし、被除数が負であれば、除数を加算し続けることで負側から0に近づける操作になります。こうなると、条件に応じて加算と減算を使い分ける必要がでてくるわけで、アルゴリズムは複雑になります。ここでは、除算の基本を解説するのが目的なので、除数、被除数ともに正であると限定して話を進めましょう。

## 4-3 回復法をもとに回路を設計してみる

では、さっそく回復法による除算回路を設計してみましょう。とりあえず、被除数を16ビット、除数を8ビットの正の整数とします。 $pr_{initial} < 2d$ とするための方法としては、回路を簡略化するため、被除数の上位に0を拡張することにします。

### ● リアルタイム性が低ければ基本的な回路構成で十分

図4-6にブロック図を示します。なお、この図にはシーケンスにかかわる信号などは表記されていないので注意してください。この回路の動作は、以下のようになります。

- 1)  $re$  (リセット・イネーブル) でシーケンスを初期化する。 $i = 0$ となり、 $start$ パルス待ちになる
- 2)  $dr$  を除数 $d$ の、 $zr$  を被除数 $z$ の格納レジスタとする。レジスタ $zr$ は上位に0拡張を行うため7ビット余分にけたをとるが、部分余の格納にも流用するので合計8ビット上位に拡張する。 $start$ パルスで $z$ 、 $d$ がそれぞれのレジスタに格納され、除算の初期化が終わる。 $z$ は、レジスタ $zr$ の低位側に格納される。また、 $i = 16$ にセットされる
- 3)  $pr - d$ が演算され、結果の符合によって商(部分商; partial quotient, 以下 $pq$ と表記)が決定され、このサイクルにおける部分余( $pr$ )が選択される。次のサイクルに移行するにあたり、 $pq$ (1ビット)はレジスタ $zr$ のLSB (least significant bit) に、 $pr$ (8ビット)はレジスタ $zr$ の上位8ビットに戻される。さらに、 $zr[14:0]$ に格納されていた未使用の $z$ は、1ビットだけシフト・アップして、 $zr[14]$ は次の $pr$ のLSBとなる。ここで、 $i = i - 1$ が実行される




図4-6 回復法による除算回路

drは除数 $d$ を、zrは被除数 $z$ を格納するレジスタ。レジスタzrは上位に0拡張を行うため7ビット余分にけたをとるが、部分余の格納にも流用するので合計8ビット上位に拡張する。なお、この図にはシーケンスにかかわる信号などは表記されていない

- 4) 3)のサイクルを繰り返すことで、レジスタzr上の $z$ のデータは上位から1ビットずつ消化され、レジスタzr上から消えていく。代わりに、レジスタzrのLSBに順次格納された $pq$ が $z$ のシフト・アップといっしょに上位へシフトされ、レジスタzrの下位側に蓄積される
- 5)  $i = 0$ となったところで除算終了となる。zr[23:16]には余( $r$ )が、zr[15:0]には商( $q$ )が残る、 $z$ はすべてレジスタ上からなくなる。合計17サイクル(17クロック)で終了となる(16サイクルより1サイクル多いのは、 $z$ と $d$ をレジスタにロードするサイクルが必要だから)

この回路構成によるVerilog HDLのRTL記述を、リスト4-1に示します。

ここでは、減算と部分余を演算する系を1系統だけ使い、繰り返しのよって除算を成立させました。リアルタイム性が必要なく、スループットが足りている場合であれば、この構成で十分です。

クロックの動作周波数が高い場合は、減算器の伝播遅延がネックになるでしょう。とくに、除数のビット幅が大きい場合、遅延が大きくなると同時に、符号ビットがドライブするマルチプレクサの制御のファンアウトも気になる場所です。なにしろ、減算器のもっとも遅いパスである符号ビットでマルチプレクサを制御するわけですから…(このあたりの改善法は、後述の非回復法で検討する)。

### ● スループットを上げたいときはパイプライン化

上述した方法は、除数が同じである限り、減算器とマルチプレクサの部分は変わらないので、被除数が大きくなっても動作周波数は変化しません。回路規模は小さくなるのですが、サイクル数(スループット)が増えることとなります。スループットを上げたい、あるいは毎クロック結果が欲しい場合には向きません。

そのような要求がなされたときのために、図4-6の構成にパイプライン化を施してみましよう。実はとても簡単に実現できるのです。

図4-7にその構成の概要を示します。図4-6の構成のままサブモジュール化し、レジスタzrにフィー

リスト4-1 図4-6のソース・コード

演算中を示すbusyを出力してある

```

module div_restore_a (
  q    , // 商 [15:00]:integer
  r    , // 余 [07:00]:integer
  busy , // busy

  z    , // 被除数 [15:00]:integer
  d    , // 除数 [07:00]:integer
  start , // startパルス (除算器トリガ)
  clk  , // クロック
  re   ); // リセット・イネーブル

// io
input [15:00] z    ;
input [07:00] d    ;
input        start ;
input        clk  ;
input        re   ;

output [15:00] q    ;
output [07:00] r    ;
output        busy ;

// reg
reg [04:00] i    ;
reg [23:00] zr   ;
reg [07:00] dr   ;

// sequence(シーケンサ)
wire en_seq = ( i > 0 ) ;

always @( posedge clk ) begin
  if      ( re )      i  <= 5' h00 ;
  else if ( start )  i  <= 5' h10 ;
  else if ( en_seq ) i  <= i - 1' b1 ;
end

// function(演算処理部)
wire [08:00] pr = zr[23:15] ; // 前サイクルのシフト・アップ済み部分余
wire [08:00] sb = pr - {1'b0,dr}; // 部分余-除数
wire        pq = ~sb[08] ; // 減算結果の符号を反転し部分商とする
wire [07:00] mx = ( sb[08] )? pr[07:00] :sb[07:00] ;
wire [23:00] zw = {mx,zr[14:00],pq} ; // 減算前と後を減算結果により選択
// 次サイクル用 zr レジスタ内容
// シフト・アップ済み部分余
// シフト・アップ済み被除数の未使用部
// 部分商

always @( posedge clk ) begin
  if      ( start ) zr  <= {8' h00,z} ;
  else if ( en_seq ) zr  <= zw ;
end

always @( posedge clk ) begin
  if      ( start ) dr  <= d    ;
end

assign q    = zr[15:00] ;
assign r    = zr[23:16] ;
assign busy = en_seq ;

endmodule

```




図4-7 回復法除算器のパイプライン化

図4-6の構成のままサブモジュール化し、レジスタ  $zr$  にフィードバックしていた  $pr$  と  $pq$  を、次のサイクル用のサブモジュールにあるレジスタ  $zr$  に渡すだけでよい

ドバックしていた  $pr$  と  $pq$  を、次のサイクル用のサブモジュールにあるレジスタ  $zr$  に渡すだけでよいのです。パイプライン化のため、 $d$  も次段に渡します。

この構造を16段繰り返すだけでパイプライン化は終了となります。必要に応じて出力段にレジスタを追加してください。 $z$  および  $pq$  をシフト・レジスタにてシフト・アップしていた部分は、モジュール間の接続によるシフトとなり、シーケンサも必要なくなります。これによってスループットを1とすることができます。

この回路構成による Verilog HDL のRTL 記述を、リスト4-2に示します。出力レジスタを付けなければ、レイテンシは16となります。

## ● 2より大きい数を法とする除算でレイテンシを抑える

図4-1と図4-2を比較すると明らかなのですが、10を法とした場合に2サイクルで終わった除算が、2を法とした場合には5サイクルかかっています。法として選択された数が大きくなると、除算を速く収束させることができるわけです。結果を得るまでのクロック数あるいはレイテンシを小さくしたい場合は効果的な方法です。

ここでは、2より大きい数を法とする除算の構成法を紹介しましょう。

2より大きい数を法とするとはいえ、3とか7を法として演算するのは論理回路の観点から適当ではありません。現状では、基本的に論理回路はバイナリ表現なので、4や8などの2のべき乗以外を法とすることは `un_filled_code` の生成につながり、効率が悪くなります。また、除算結果が3や7を法とした表現では周辺回路との整合が悪く、扱いもめんどろになります。ここでは、4を法として選択することになります。

では、図4-1の筆算を、4を法とした表現に変換するところから始めましょう。

図4-8(a)は、 $152 \div 11$  を4を法として演算しています。これはこれで正しい演算なのですが、1けた

リスト4-2 パイプラインを施した回路

```

module div_restore_b (
    q , // 商 [15:00]:integer
    r , // 余 [07:00]:integer

    z , // 被除数 [15:00]:integer
    d , // 除数 [07:00]:integer
    clk); // クロック

// io
input [15:00] z ;
input [07:00] d ;
input clk ;

output [15:00] q ;
output [07:00] r ;

// wire
wire [23:00] zi_15 , zi_14 , zi_13 , zi_12 , zi_11 , zi_10 , zi_09 , zi_08 ;
wire [23:00] zi_07 , zi_06 , zi_05 , zi_04 , zi_03 , zi_02 , zi_01 , zi_00 ;
wire [07:00] di_15 , di_14 , di_13 , di_12 , di_11 , di_10 , di_09 , di_08 ;
wire [07:00] di_07 , di_06 , di_05 , di_04 , di_03 , di_02 , di_01 , di_00 ;
wire [23:00] zo_00 ;
wire [07:00] do_00 ;

// function
assign zi_15 = {8'h00,z} ;
assign di_15 = d ;

div_restore_b_sub u15 (
    .zo (zi_14) , .do(di_14) ,
    .zi (zi_15) , .di(di_15) , .clk(clk) );
div_restore_b_sub u14 (
    .zo (zi_13) , .do(di_13) ,
    .zi (zi_14) , .di(di_14) , .clk(clk) );
div_restore_b_sub u13 (
    .zo (zi_12) , .do(di_12) ,
    .zi (zi_13) , .di(di_13) , .clk(clk) );
div_restore_b_sub u12 (
    .zo (zi_11) , .do(di_11) ,
    .zi (zi_12) , .di(di_12) , .clk(clk) );
div_restore_b_sub u11 (
    .zo (zi_10) , .do(di_10) ,
    .zi (zi_11) , .di(di_11) , .clk(clk) );
div_restore_b_sub u10 (
    .zo (zi_09) , .do(di_09) ,
    .zi (zi_10) , .di(di_10) , .clk(clk) );
div_restore_b_sub u09 (
    .zo (zi_08) , .do(di_08) ,
    .zi (zi_09) , .di(di_09) , .clk(clk) );
div_restore_b_sub u08 (
    .zo (zi_07) , .do(di_07) ,
    .zi (zi_08) , .di(di_08) , .clk(clk) );
div_restore_b_sub u07 (
    .zo (zi_06) , .do(di_06) ,
    .zi (zi_07) , .di(di_07) , .clk(clk) );
div_restore_b_sub u06 (
    .zo (zi_05) , .do(di_05) ,
    .zi (zi_06) , .di(di_06) , .clk(clk) );
div_restore_b_sub u05 (
    .zo (zi_04) , .do(di_04) ,
    .zi (zi_05) , .di(di_05) , .clk(clk) );
div_restore_b_sub u04 (
    .zo (zi_03) , .do(di_03) ,

```